
Modeling and Specification 

Topics

What is QLOCK?

Modeling and Description of QLOCK in OTS

Formal specification of QLOCK in CafeOBJ

Formal specification of mutual exclusion

Modeling, Specifying, and Verifying (MSV)
in CafeOBJ

1. By understanding a problem to be 
modeled/specified, determine several sorts of 
objects (entities, data, agents, states) and 
operations (functions, actions, events) over 
them for describing the problem

2. Define the meanings/functions of the 
operations by declaring equations over 
expressions/terms composed of the operations

3. Write proof scores for properties to be verified

MSV with proof scores in CafeOBJ

Understand problem 
and construct model

Write system spec SPsys and
Write property spec SPprop

Construct proof score of 
SPprop w.r.t. SPsys

An example: mutual exclusion protocol

Assume that many agents (or processes) are 
competing for a common equipment, but at 
any moment of time only one agent can use 
the equipment.  That is, the agents are 
mutually excluded in using the equipment.  A 
protocol (mechanism or algorithm) which can 
achieve the mutual exclusion is called “mutual 
exclusion protocol”.

QLOCK (locking with queue): 
a mutual exclusion protocol 

Remainder Section

Critical Section

Is i at the top 
of  the queue?

cs

Put its name i into the 
bottom of the queue

Remove/get the 
top of the queue

wt

rm

true

false

Each agent i is executing: : atomic action

QLOCK: basic assumptions/characteristics

There is only one queue and all agents/processes 
share the queue.
Any basic action on the queue is inseparable (or 
atomic).  That is, when any action is executed on the 
queue, no other action can be executed until the 
current action is finished.
There may be unbounded number of agents.
In the initial state, every agents are in the remainder 
section (or at the label rm), and the queue is empty.

The property to be shown is that at most one agent 
is in the critical section (or at the label cs) at any 
moment.

Global (or macro) view of QLOCK

…k j i

i

k

j

is i?

is j?
put

get

get

…

: queue

: agents
put

Modeling QLOCK (via Signature Diagram) 
with OTS (Observational Transition System)

…
kj i

i

k

j

is i?

is j?
put

get

get

…

put

Queue

Label

Pid

Sys

want

try

pc

queue

exit

init

Signature for QLOCKwithOTS

Sys is the sort for representing the state space of the system.
Pid is the sort for the set of agent/process names.
Label is the sort for the set of labels; i.e. {rm, wt, cs}.
Queue is the sort for the queues of Pid
pc (program counter) is an observer returning a label where 
each agent resides.
queue is an observer returning the current value of the 
waiting queue of Pid.
want is an action for agent i of putting its name/id into the 
queue.
try is an action for agent i of checking whether its name/id 
is at the top of the queue.
exit is an action for agent i of removing/getting its name/id 
from the top of the queue.

Observation declaration

action declaration

visible sort declaration

Hiden sort declaration

CafeOBJ signature for QLOCKwithOTS

-- state space of the system
*[Sys]*

-- visible sorts for observation
[Queue Pid Label] 

-- observations 
bop pc : Sys Pid -> Label
bop queue : Sys -> Queue

-- actions
bop want : Sys Pid -> Sys
bop try  : Sys Pid -> Sys
bop exit : Sys Pid -> Sys

Module LABEL specifying 
(via tight denotation/semantics) “labels”

mod! LABEL {
[Label]
ops rm wt cs : -> Label
pred (_=_) : Label Label {comm}
var L : Label
eq (L = L) = true .
eq (rm = wt) = false .
eq (rm = cs) = false .
eq (wt = cs) = false .

}

Predicate (_ = _) defines identity relation 
among rm, wt, and cs.

qlock.mod

Module PID specifying (via loose denotation) 
“agent/process names/identifiers”

mod* PID {
[Pid < PidErr]
op none : -> PidErr
pred (_=_) : PidErr PidErr {comm}
var I : Pid .
eq (I = I) = true .
eq (none = I) = false .
-- (none = none) is not defined intentionally 

}
The constant none of  the sort PidErr is intended to 
indicate the result of getting top of the empty queue.
Any element in the sort Pid is defined not equal to none, 
that is, return false for predicate (_ = _) .
Notice that (none = none) does not reduced to true or 
false.

qlock.mod

Module QUEUE specifying “queue” (1)

-- an parameterized module

mod* TRIVerr {
[Elt < EltErr]
op none : -> EltErr

}
mod! QUEUE(D :: TRIVerr) {
[Queue]

-- constructors
op empty :             -> Queue {constr}
op _,_   : Queue Elt.D -> Queue {constr l-assoc}

-- operators
op put : Elt.D Queue -> Queue
op get : Queue -> Queue
op top : Queue -> EltErr.D
op empty? : Queue -> Bool

qlock.mod

Module QUEUE specifying “queue” (2)

-- an parameterized module

-- CafeOBJ variables
var Q : Queue
vars X Y : Elt.D

-- equations
eq put(X,empty) = empty,X .
eq put(X,(Q,Y)) = put(X,Q),Y .
-- get(empty) is not defined intentionally
eq get((Q,X)) = Q .
eq top(empty) = (none):EltErr.D .
eq top((Q,X)) = X .
eq empty?(empty) = true .
eq empty?((Q,X)) = false .

}

qlock.mod

QLOCK using operators 
in the CafeOBJ module QUEUE

Remainder Section

Critical Section

top(queue)=i

cs

put(queue,i)

get(queue)

wt

rm
true

false

Each agent i is executing: : atomic action

want

try

exit

Module QLOCK specifying “QLOCK” (1-1)

view TRIVerr2PID from TRIVerr to PID {
sort Elt -> Pid,
sort EltErr -> PidErr,
op (none):EltErr -> (none):PidErr }

mod* QLOCK {
pr(LABEL)
pr(QUEUE(D <= TRIVerr2PID))
*[Sys]*

-- any initial state
op init : -> Sys

-- observations
bop pc    : Sys Pid -> Label
bop queue : Sys -> Queue

-- actions
bop want : Sys Pid -> Sys
bop try  : Sys Pid -> Sys
bop exit : Sys Pid -> Sys

-- for any initial state
eq pc(init,I:Pid)  = rm .
eq queue(init) = empty .

qlock.mod

Module QLOCK specifying “QLOCK” (1-2)

mod* QLOCK {
pr(LABEL)
pr(QUEUE(PID{sort Elt -> Pid,

sort EltErr -> PidErr,
op (none):EltErr -> none):PidErr}))

*[Sys]*
-- any initial state

op init : -> Sys
-- observations

bop pc    : Sys Pid -> Label
bop queue : Sys -> Queue

-- actions
bop want : Sys Pid -> Sys
bop try  : Sys Pid -> Sys
bop exit : Sys Pid -> Sys

-- for any initial state
eq pc(init,I:Pid)  = rm .
eq queue(init) = empty .

qlock.mod

Module QLOCK specifying “QLOCK” (2)

var S : Sys . vars I J : Pid .
-- for want
op c-want : Sys Pid -> Bool {strat: (0 1 2)}
eq c-want(S,I) = (pc(S,I) = rm) .
--
ceq pc(want(S,I),J)  

= (if I = J then wt else pc(S,J) fi) 
if c-want(S,I) .

ceq queue(want(S,I)) = put(I,queue(S))   
if c-want(S,I) .

ceq want(S,I) = S                 
if not c-want(S,I) .

qlock.mod

Module QLOCK specifying “QLOCK” (3)

-- for try
op c-try : Sys Pid -> Bool {strat: (0 1 2)}
eq c-try(S,I) = (pc(S,I) = wt and top(queue(S)) = I) .
--
ceq pc(try(S,I),J)  

= (if I = J then cs else pc(S,J) fi) if c-try(S,I) .
eq queue(try(S,I)) = queue(S) .
ceq try(S,I)        = S                  if not c-try(S,I) .

-- for exit
op c-exit : Sys Pid -> Bool {strat: (0 1 2)}
eq c-exit(S,I) = (pc(S,I) = cs) .
--
ceq pc(exit(S,I),J)  

= (if I = J then rm else pc(S,J) fi) if c-exit(S,I) .
ceq queue(exit(S,I)) = get(queue(S))     if c-exit(S,I) .
ceq exit(S,I)        = S                 if not c-exit(S,I) .

}

qlock.mod

(_ =*= _) is congruent for OTS

The binary relation (S1:Sys =*= S2:Sys) is defined to 
be true iff S1 and S2 have the same observation values.

OTS style of defining the possible changes of the values of 
obervations is characterized by the equations of the form:
o(a(s,d),d’) = ...o1(s,d1)...o2(s,d2)...on(s,dn)...
for appropriate data values of d,d’,d1,d2,...,dn .

It can be shown that OTS style guarantees 
that (_ =*= _) is congruent with respect 
to all actions.

RQLOCK (set of reachable states) of 
OTSQLOCK (OTS defined by the module QLOCK)

-- any initial state
op init : -> Sys

-- actions
bop want : Sys Pid -> Sys
bop try  : Sys Pid -> Sys
bop exit : Sys Pid -> Sys

Signature determining RQLOCK

RQLOCK = {init} ∪
{want(s,i)|s∈RQLOCK,i∈Pid} ∪
{try(s,i) |s∈RQLOCK,i∈Pid} ∪
{exit(s,i)|s∈RQLOCK,i∈Pid} 

Recursive definition of RQLOCK

Mutual exclusion property 
as an invariant

mod INV1 {
pr(QLOCK)

-- declare a predicate to verify to be an invariant
pred inv1 : Sys Pid Pid

-- CafeOBJ variables
var S :  Sys .
vars I J : Pid . 

-- define inv1 to be the mutual exclusion property
eq inv1(S,I,J) 

= (((pc(S,I) = cs) and (pc(S,J) = cs)) implies I = J) .
}

INV1 |= ∀s∈RQLOCK∀i,j∈Pid.inv1(s,i,j)
Formulation of proof goal for mutual exclusion property

invariants-0.mod

Induction scheme 
induced by the structure of RQLOCK

mx(s) =def ∀i,j∈Pid.inv1(s,i,j) 

{              INV1 |= mx(init),
INV1∪{mx(s)=true} |= ∀k . mx(want(s,k)),
INV1∪{mx(s)=true} |= ∀k . mx(try(s,k)),
INV1∪{mx(s)=true} |= ∀k . mx(exit(s,k)) }    

implies
INV1 |= ∀s∈RQLOCK.mx(s)

Schematic signature diagram for OTS

Hidden Sort
(State Space)

Visible Sort
(Data)

．．．

．．．

Action
(method)

Action
(method)

Observation
(attribute)

Observation
(attribute)

Visible Sorts
(Data)

Visible Sorts
(Data)

Visible Sort
(Data)

．．．

VSs

HSs
Coherent


